


Beispielprogramme für MDRIVE / MFORCE Plus Motion Control (RS485)

Im Folgenden werden 4 ASCII-Beispielprogramme mit Funktion und Programm-Syntax kurz beschrieben, um die umfangreiche Funktionalität und die einfache Programmierung der

- **MDRIVE Plus Motion Control** – Schrittmotorantriebe und
- **MFORCE Motion Control** – Schrittmotorsteuerungen

mit RS485-Schnittstelle und digitalen Ein-/Ausgängen (IOs) zu demonstrieren.

Die auskommentierten Programme sind farbig so dargestellt, wie diese im Editor-Fenster des IMS-Terminal-Programmes mit der automatischen Befehls- und Syntaxerkennung erscheinen.



Im Editor-Fenster werden Kommentare nach dem Hochkomma grün dargestellt und beim Download ignoriert, wenn pro Zeile nicht mehr als insgesamt 64 Zeichen verwendet werden !

Vor dem Download empfiehlt es sich mit dem Befehl „fd“ die jeweilige Steuerung auf Werkseinstellungen zurückzusetzen und alte Programme zu löschen.

Folgende Beispielprogramme:

1. Einfache Positionierung (1 Umdrehung)
2. Referenzfahrt auf Sensor, Absolut- und Relativpositionierung, selbststartend
3. Drehzahlsteuerung von 0 bis 4 Umdr./s über Analog-Eingang 0 bis 10V
4. Encoderbetrieb, Positionierung 0 bis 360° über Eingänge 1 bis 3, Fehlerbehandlung

Für eine einwandfreie Funktion wird ein ordnungsgemäßer Anschluss und Betrieb der jeweiligen Antriebe und Steuerungen gemäß den Hard- und Software-Dokumentationen vorausgesetzt, die unter www.kocomotion.de auf den jeweiligen Produktseiten zum Download zur Verfügung stehen.

1. Einfache Positionierung (1 Umdrehung)

Es wird der Positionszähler auf Null gesetzt und eine Relativ-Positionierung von genau 1 Umdrehung mit 50.000 Mikroschritten ausgeführt.

Das Programm wird gestartet durch die Eingabe des Befehls „EX xx“ .

'Einstellungen

Hc=5 'Haltestrom 5%
Rc=40 'Laufstrom 40%
Ms=250 'Auflösung 250 Mikroschritte (MS), d.h. 50.000 MS/Umdr.
Vi=5000 'Startdrehz. 5.000 MS/s = 0,1 Umdr./s
Vm=200000 'Enddrehz. 200.000 MS/s = 4 Umdr./s
A=400000 'Beschl. 400.000 MS/s² = 8 Umdr./s²
D=A 'Bremsrampe = Beschleunigung

'Programm

PG 1 'Start Programm-Mode ab Adresse 1

LB xx 'Name/Label "xx"
P=0 'Positionszähler auf 0
MR 50000 'Bewegung relativ 50.000 Schritte (1 Umdr.)
H 'warten bis Bewegung fertig
E 'Programmende

PG 'Beenden Programm-Mode
S 'Abspeichern im EEPROM nach Download

2. Referenzfahrt auf Sensor, Absolut- und Relativpositionierung, selbststartend

IO 1 wird als Referenz-Eingang (Home) mit npn-Verhalten definiert, so dass ein einfacher Schließer nach Masse (GND) genügt.

IO 4 wird als Bewegungs-Ausgang (Moving) mit npn-Verhalten definiert, so dass dieser immer dann aktiv ist und nach Masse schaltet, wenn sich der Motor bewegt.

Das Programm führt eine Referenzfahrt auf IO 1 aus, setzt den Positionszähler auf Null und führt dann eine Absolut-Positionierung von genau 1 Umdrehung aus und fährt dann relativ 0,2 Umdrehungen zurück.

Vor und nach den Positionierungen wird der Wert des Positionszählers auf die Schnittstelle RS485 ausgegeben und erscheint z.B. im Terminal-Fenster des IMS-Terminal-Programmes.

Das Programm wird gestartet durch die Eingabe des Befehls „EX su“. Der Name „su“ bedeutet „Startup“ und ist der Autostart-Name, so dass das Programm ab dieser Adresse auch ohne Start-Befehl und nur nach dem Zuschalten der jeweiligen Versorgungsspannung startet.

'Einstellungen

Hc=30 'Haltestrom 30%
Rc=60 'Laufstrom 60%
Ms=250 'Auflösung 250 Mikroschritte (MS), d.h. 50.000 MS/Umdr.
Vi=5000 'Start-Drehzahl 5.000 MS/s = 0,1 Umdr./s
Vm=400000 'End-Drehzahl 400.000 MS/s = 8 Umdr./s
A=1000000 'Beschleunigung 1.000.000 MS/s² = 20 Umdr./s²
D=A 'Bremsrampe = Beschleunigung
S1=1,0,0 'IO1 als Home-Eingang, low aktiv, npn
S4=17,0,0 'IO4 als Moving-Ausgang, low aktiv, npn

```

'Programm
PG 1      'Start Programm-Mode ab Adresse 1

LB su     'Name/Label "su" – Startup / Autostart
HM 1     'Referenzfahrt auf Home-Schalter
H        'Warten bis Beweg. beendet
P=0      'Positionszähler auf 0 setzen
PR "P=",P 'Print / Ausgabe Position auf RS485
MA 50000 'Absolutbew. 50.000 Schritte = 1 Umdr.
H        'Warten bis Beweg. beendet
PR "P=",P 'Print / Ausgabe Position auf RS485
MR -10000 'Relativbew. -10.000 Schritte = -0,2 Umdr.
H        'Warten bis Beweg. beendet
PR "P=",P 'Print / Ausgabe Position auf RS485
E        'Programmende

PG       'Beenden Programm-Mode
S        'Abspeichern im EEPROM nach Download

```

3. Drehzahlsteuerung von 0 bis 4 Umdr./s über Analog-Eingang 0 bis 10V

Die ersten 3 Eingänge IO 1 bis 3 werden als Eingänge mit npn-Verhalten definiert:

- Eingang 1 als GO-Eingang / Taster, der bei Betätigung das im EEPROM abgespeicherte Programm ab der Adresse 1 startet
- Eingang 2 als STOP-Eingang / Taster, der bei Betätigung das laufende Programm und die Bewegung abbricht
- Eingang 3 wird im Programm als Richtungs-Eingang verwendet

IO 4 wird als Bewegungs-Ausgang (Moving) mit npn-Verhalten definiert, so dass dieser immer dann aktiv ist und nach Masse schaltet, wenn sich der Motor bewegt.

Das Programm arbeitet in einer Endlosschleife und fragt den Analogeingang ab, multipliziert den Wert mit 200 und erzeugt mit dem SL-Befehl die entsprechende Drehzahl von maximal ca. 4 Umdrehungen pro Sekunde.

Eingang 3 wird ebenfalls kontinuierlich abgefragt und bestimmt die Drehrichtung.

Das Programm wird gestartet durch die Eingabe des Befehls „EX xx“ oder durch Schalten des Eingangs 1 (GO-Eingang).

```

'Einstellungen
Hc=20     'Haltestrom 20%
Rc=50     'Laufstrom 50%
Ms=250    'Auflösung 250 Mikroschritte (MS), d.h. 50.000 MS/Umdr.
Vi=5000   'Start-Drehzahl      5.000 MS/s = 0,1 Umdr./s
Vm=500000 'End-Drehzahl      500.000 MS/s = 10 Umdr./s
A=2500000 'Beschleunigung 2.500.000 MS/s² = 50 Umdr./s²
D=A       'Bremsrampe = Beschleunigung
S1=4,0,0 'IO1 als GO-Eingang (Adr. 1), low-aktiv, npn
S2=5,0,0 'IO2 als STOP-Eingang, low-aktiv, npn
S3=0,0,0 'IO3 als allg. Eingang, low-aktiv, npn
S4=17,0,0 'IO4 als Moving-Ausgang, low-aktiv, npn
S5=9,1    'Analog-Eingang, Spannung: 0...10 V
D1=10     'Digitalfilter für Eingang 1 -> 10ms
D2=10     'Digitalfilter für Eingang 2 -> 10ms
D3=10     'Digitalfilter für Eingang 3 -> 10ms
D5=10     'Filter für Analog-Eingang
Mt=50     'Motorausschwingzeit 50ms

```

```

'Programm
PG 1      'Start Programm-Mode Adresse 1

LB xx     'Name/Label "xx"
LB x1     'Name/Label "x1" - Endlosschleife
  R1=I5   'Analogwert in Register R1 einlesen
  R1=R1*200 ' *200 (max. 1023*200 -> max. ca. 4 Umdr./s)
  BR x2,I3=1 'Sprung zu "x2" wenn Richtungs-Eing. I3=1
  SL R1    'Drehen im Uhrzeigersinn mit Wert aus R1
  BR x1    'Sprung zu "x1", Endlosschleife
  LB x2    'Name/Label "x2" - andere Drehrichtung
  SL -R1   'Drehen entgegen Uhrzeigersinn mit Wert aus R1
  BR x1    'Sprung zu "x1", Endlosschleife
E         'Programmende

PG        'Ende Programm-Mode
S         'Abspeichern im EEPROM nach Download

```

4. Encoderbetrieb, Positionierung 0 bis 360° über Eingänge 1 bis 3, Fehlerbehandlung

Wenn ein integrierter Encoder vorhanden ist, dann können wie in diesem Beispiel mit dem Befehl EE=1 die Encoderfunktionen eingeschaltet werden, so dass dann mit einer festen Auflösung von 2048 Encoder-Schritten pro Umdrehung gearbeitet wird.

IO 4 wird als Schleppfehler-Ausgang (Stall) mit npn-Verhalten definiert. Dieser ist dann aktiv, wenn ein Schleppfehler und damit ein Überlastfall auftritt.

Die ersten 3 Eingänge IO 1 bis 3 werden als normale Eingänge mit npn-Verhalten definiert, so dass jeweils ein einfacher Schließer nach Masse (GND) genügt.

Das Programm arbeitet in einer Endlosschleife und fragt den Zustand der Eingänge 1 bis 3 ab und positioniert den Motor in Abhängigkeit davon absolut auf 0°, 90°, 180°, 270° oder 360°. Dabei werden die Eingänge 1 bis 3 als 3-bit-Wort eingelesen und dann entsprechende Unterprogramme aufgerufen.

Über die Interrupt-Funktion „OE“ (Error-Handler) wird eine Fehlerbehandlungs-Routine aktiviert, die im Fehlerfall ausgeführt wird. Dabei wird der entsprechende Fehler über die RS485 gemeldet und nach 3 Sekunden zurückgesetzt.

Das Programm wird gestartet durch die Eingabe des Befehls „EX su“. Der Name „su“ bedeutet „Startup“ und ist der Autostart-Name, so dass das Programm ab dieser Adresse auch ohne Start-Befehl und nur nach dem Zuschalten der jeweiligen Versorgungsspannung startet.

```

'Einstellungen
Ee=1      'Encoder ein, also 2048 Encoder-Schritte/Umdr.
Vm=4096   'End-Drehzahl 4.096 Enc.-Schritte/s = 2 Umdr./s
Vi=200    'Start-Drehzahl 200 Enc.-Schritte/s = 0,1 Umdr./s
A=20480   'Beschleunigung 20.480 Enc.-Schritte/s² = 10 Umdr./s²
D=A       'Bremsrampe = Beschleunigung
Hc=50     'Haltestrom 50%
Rc=60     'Laufstrom 60%
Mt=50     'Motor-Ausschwingzeit 50 ms
Sf=30     'zul. Schleppfehler max. +/-30 Enc.-Schritte
Sm=0      'Stop wenn Schleppfehler erkannt
Db=2      'zul. Toleranz am Bew.-ende max. +/-2 Enc.-Schritte
Pm=0      'Positionsnachregelung am Bew.-ende aus
S1=0,0,0 'IO1 als normaler Eingang, low-aktiv, npn
S2=0,0,0 'IO2 als normaler Eingang, low-aktiv, npn
S3=0,0,0 'IO3 als normaler Eingang, low-aktiv, npn
S4=19,0,0 'IO4 als Schleppfehler-Ausgang, low-aktiv, npn

```

```

'Programm
PG 1      'Start Programm-Modus ab Adresse 1

LB su     'Name/Label "su" – Startup / Autostart
OE yy     'Error-Handler "yy" aktivieren
P=0      'Positionszähler auf Null
LB y1     'Name/Label "y1" - Endlosschleife
R1=ln     'Einlesen I1 bis I4 als 4 Bit-Wort in Variable R1
R1=r1&7   'bitweiser Vergleich mit 7 (I4 ignorieren)
CL x1,R1=7 'Unterprogramm "x1" -> 0°      I1/2/3: 0,0,0
CL x2,R1=6 'Unterprogramm "x2" -> 90°     I1/2/3: 1,0,0
CL x3,R1=5 'Unterprogramm "x3" -> 180°    I1/2/3: 0,1,0
CL x4,R1=4 'Unterprogramm "x4" -> 270°    I1/2/3: 1,1,0
CL x5,R1=3 'Unterprogramm "x5" -> 360°    I1/2/3: 0,0,1
H 10     '10ms warten
BR y1    'Sprung zu "y1", Endlosschleife
E        'Ende Hauptprogramm

LB x1     'Name/Label "x1" -> Unterprogramm 0°
MA 0     'absolut auf 0° fahren
H        'Warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm

LB x2     'Name/Label "x2" -> Unterprogramm 90°
MA 512   'absolut auf 90° fahren
H        'Warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm

LB x3     'Name/Label "x3" -> Unterprogramm 180°
MA 1024  'absolut auf 180° fahren
H        'Warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm

LB x4     'Name/Label "x4" -> Unterprogramm 270°
MA 1536  'absolut auf 270° fahren
H        'Warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm

LB x5     'Name/Label "x5" -> Unterprogramm 360°
MA 2048  'absolut auf 360° fahren
H        'Warten bis Bew. beendet
RT       'Rücksprung aus Unterprogramm

LB yy     ' Name/Label "yy" -> Error-Handler
PR "Fehler=",Er 'Print / Ausgabe Fehler-Variable auf RS485
H 100    '100 ms warten für PR-Befehl
Er=0     'Fehler-Variable zurücksetzen
H 3000   '3 s. warten
St=0     'wenn nötig Schleppfehler-Flag zurücksetzen
RT       'Rücksprung aus Unterprogramm

PG        'Beenden Programm-Mode
S        'Abspeichern im EEPROM nach Download

```

Änderungen vorbehalten / Rev. 09/2011



KOCO
MOTION

... Intelligence in motion

KOCO MOTION GmbH
Niedereschacher Straße 54
78083 Dauchingen
Germany

Telefon +49 7720 995858-0
Telefax +49 7720 995858-9
E-Mail: info@kocomotion.de
www.kocomotion.de

Technical Support:
Telefon +49 7720 995858-3
Telefax +49 7720 995858-9
E-Mail: m.weber@kocomotion.de

Sales/Technical Support:
Telefon +49 35205 4587-8
Telefax +49 35205 4587-9
E-Mail: h.ruhland@kocomotion.de